



● 微计算机程序库丛书

BASIC

实用程序100例

赵旭 冯克久 编著

电子工业出版社

BASIC 实用程序100例

赵旭 冯克久 编

電子工業出版社

内 容 提 要

本书是根据国内外各种实际应用问题和有关资料编制的实用程序集。内容涉及到数理统计、回归分析、微积分计算、高次方程和方程组求解、矢量及矩阵运算、绘制(打印)函数的图形、有关投资、贷款、储蓄、折旧处理和债务偿还、线性规划、通用报表打印以及其它实用程序等。可用于工业、农业、教育、银行信贷、实验研究及社会科学等领域。

全书分十章,共102个实用程序。均用 BASIC语言写成。每个程序中的各个具有独立功能的段落、数据输入方式等都用 REM 说明语句给以扼要标注性说明。具有人机对话功能和简明易懂、便于修改、调试等特点。由于大、中、小和微型计算机都配有 BASIC语言,所以本程序具有适用机种广、应用领域大等特点。

本书适于计算机应用人员、工程技术人员、各级学校师生和计算机爱好者参考、使用。

本书配有软盘可供读者选购。

BASIC实用程序100例

赵旭 冯克久编

责任编辑 王小民

电子工业出版社出版

(北京市万寿路)

*

山东电子工业印刷厂印刷

新华书店北京发行所发行 各地新华书店经售

*

开本: 787 × 1092 1/16 印张: 15 字数: 349 千字

1985年4月第1版 1985年5月第1次印刷

印数: 44700册 定价: 2.70元

统一书号: 15290·170

序 言

BASIC-程序100例是根据国内外各种实际应用问题和有关资料文献编制的实用程序汇集。

内容涉及到数理统计（数据整理、统计分布及检验）、回归分析、微积分计算、高次方程和方程组求解、矢量及矩阵运算、绘制（打印）函数的轨迹、有关投资、贷款、储蓄、折旧处理和债务偿还以及线性规划、通用报表设计处理和其它实用程序等。可用于工业、农业、实验科学以及教育、银行信贷、社会科学等领域。

全书分十章，共有102个实用程序。均用 BASIC 语言写成。每个程序中的各个具有独立功能的段落、数据输入方式等均用 REM 说明语句进行扼要标注性说明。简单易懂，具有人机对话功能和便于修改、调试等特点，由于国内外各种大、中、小和微型电子计算机多配有 BASIC 语言，所以本程序具有适用机种多及应用领域广等特点。

本书的全部程序已在 CROMEMCO 等微机上运行、通过。为方便读者了解、使用程序，对全部程序均给出运行实例。

102个程序的全部程序量约240K（字节），每个实用程序都具有独立的功能、可单独运行。也可由使用者把几个有关的程序组合在一起使其具有更多、更强的功能；或是把其中一个或几个程序稍加修改（变动数据输入方式，加 RETURN 语句）作为用户程序的子程序供读者随时调用。

国外软件工程师中流行这样一句话，“充分利用已有的一切程序，那怕只有 3 或 5 条语句也好。”，可省去不必要的重复劳动。集累已有的功能程序可使你左右逢源、得心应手，无疑会加速程序编制的速度、促进程序设计积木化、结构化的进程。

全部程序均采用统一格式编写，依次为：程序标题、程序说明、程序正文和运行实例四部分。为方便读者了解和使用本程序，在程序说明部分还给出了该程序功能。程序中使用的主要符号（参数、变量）说明、程序设计（计算）的理论依据和程序在磁盘上的存储名称。

程序在磁盘上的存储均以 ASCII 码方式存储，便于程序在内存中的合并（MERGE 或 CHAIN）。

下面按程序的分类给以扼要说明。

（一）数理统计程序

包括15个应用程序。主要是关于统计数据的处理和一些重要频数分布。如二项式分布、普哇松分布、正态分布、 χ^2 分布、学生氏 T 分布、F 分布等以及几种分布的检验计算程序。这些程序可广泛地应用于工业、农业、卫生系统及误差分析、可靠性设计计算等领域。还把学生、运动员的成绩统计、名次排队打印等程序也放在本章。

（二）回归分析程序

包括6个实用程序，有一元线性及多元线性回归、多项式回归、几何回归和指数回归等常用回归分析程序。通过这些程序去寻求各种工农业生产、科学实验及社会科学、医学等领域的各种观察、实验、统计等数据的内在规律及其数学模型。它是一种了解、发现和掌握客观规律性的重要手段。

（三）微积分计算程序

包括5个程序。其中有3个是关于用不同的方法（辛普森法、梯形法和高斯法）求多种函数积分的计算程序，各有特点，供读者选用。此外，还给出了求导数的计算程序。

（四）高次方程及方程组求解程序

包括5个计算程序。主要给出了高次方程的牛顿法及半区间法求解程序以及多元线性方程组求解计算程序供选用。

（五）矢量和矩阵计算程序

包括5个程序。给出了常用的矢量及矩阵的加、减、点积、叉积计算程序和矩阵求逆程序。

（六）数制、单位、坐标转换及绘制函数的图形程序

包括12个程序。主要有数制转换：十六进制—十进制—二进制之间的相互转换和多种单位换算，如弧度—角度、华氏—摄氏及17种英制—公制单位换算等以及坐标系（笛卡尔坐标—极坐标）转换等基本计算程序。还给出了线性和曲线插值、由坐标值绘制（打印）其图形以及打印函数在极坐标系中的图形等程序。

（七）投资、贷款、储蓄、折旧处理和债务偿还程序

包括21个程序。其中有关投资方面的程序6个，贷款方面计算程序7个、储蓄计算程序3个、折旧处理计算程序3个以及赚得利息表和债务偿还计算程序等。

（八）线性规划计算程序

用于求解各种受线性约束的变量的线性函数的最优化计算。可广泛地用在管理科学、最优设计、运筹学等领域去追求单目标最优化。程序采用单纯形 simplex 解法。使用时只须使用者用 DATA 语句输入有关数据值和回答几个问题，即可自动计算并输出结果。

（九）通用报表设计处理程序

目前世界上对计算机汉字处理有多种形式和方案。单就国内而言，到目前已经设计了400多种。其中有40余种已在计算机上实现。目前正在抓紧编码的评测、选优工作。

由于汉字输入方式和编码的多样性，所以目前各机配备的汉字系统也是因机而异的。通用报表处理程序的汉字系统是以美国星茂（SIGMA）公司的 SIGMA-10 型微机系统为背景。

该系统的汉字采用三角编码方案，和美国王安公司相同。目前纺织部设计院电子计算机应用开发组已成功地将该汉字系统移植到国产 BCM-II 微型计算机上。

在通用报表处理程序中，汉字的显示、打印采用子程序的形式独立给出。所以，对于不同的机种所配备的不同汉字系统仅改动相应的汉字处理子程序即可使用、移植此报表处理程序。

为了方便读者使用、调试，把通用报表处理程序分拆成11个程序提供给读者。每个

程序均可独立使用、运行。由于一般报表均由标题部分、表头、表格正文以及报表封面四部分组成，所以本章分别给出了相应的实用程序。以便于单独使用、修改、调试。如把设计、调试好的数据（包括标题数据、表头数据和表格正文数据）依次仍用 DATA 语句放在通用报表打印程序中去，即可打印出完整的报表。当然也可从一开始就利用本程序设计、修改、调试，“毕其功于一役”。不过对初学者似乎有些冗长。但也是可行的。

（十）其它实用程序

包括 20 个程序。主要有未来年历打印程序、两日期间天数计算程序、查某一天是星期几、艺术图案设计处理程序。有人称做“秘书程序”。还有其它一些实用程序。简单实用、既可独立运行也可稍加修改（数据输入、输出方式）后插入到用户的应用程序之中。

如读者将全部（102 个）实用程序键入并存储在软、硬磁盘或磁带等媒体上。可构成十分有用的“实用程序库”。使用起来会更加自如。

考虑到我国的计算机应用目前正处于开始普及阶段，本书中基础应用程序，如求一元二次方程的根，解三角形、求多边形的面积、二进制—十进制—十六进制数的转换等程序也占有一定比重。

由于目前国内生产和引进的机种很多，其 BASIC 语言版本各异，如 CBASIC、MBASIC、BASIC-80、XITAN-BASIC 等。在本书各实用程序中使用语句的选择上尽量选用各机种 BASIC 文本中通用的语句。以求其适应机种广的目的。但仍会有某机种不认的语句。遇此，读者可根据该语句的功能，改成机器所认的语句即可。

本书所给出的全部程序是实际应用中的程序，不是（也达不到）最优化程序。由于笔者水平所限，书中肯定还会有应商榷、改进、甚至错误之处，请读者批评、指正。并希望在实际应用中得到完善、发展。

编者

1984.8.

目 录

第一章 数理统计程序

1. 二项式分布计算程序	2
2. 普哇松分布计算程序	3
3. 正态分布计算程序	4
4. X^2 分布计算程序	6
5. 学生氏 T 分布计算程序	7
6. F 分布计算程序	9
7. 学生氏 T 分布检验计算程序	11
8. X^2 分布检验程序	13
9. U 检验计算程序	15
10. 排列—组合计算程序	17
11. 平均值—方差—标准离差计算程序	19
12. 几何平均值—几何离差计算程序	21
13. 学生成绩统计程序—1	22
14. 学生成绩统计程序—2	24
15. 运动员成绩名次统计程序	25

第二章 回归分析计算程序

1. 一元线性回归分析程序	27
2. 多元线性回归分析程序	29
3. 指数回归分析程序	33
4. 几何回归分析程序	35
5. N 阶回归分析程序	36
6. 线性相关系数计算程序	39

第三章 微积分计算程序

1. 求导数计算程序	41
2. 求函数的积分计算程序 (辛普森法)	42

3. 梯形法定积分计算程序·····	44
4. 高斯 (GRUSS) 法定积分计算程序 ·····	45
5. 三种积分计算比较程序·····	47

第四章 高次方程及方程组求解程序

1. 求解一元二次方程的根计算程序·····	49
2. 求解多项式 (高次代数方程) 的实根计算程序 (牛顿法) ·····	50
3. 求解多项式 (高次代数方程) 在区间 $[A, B]$ 内的所有单实根计算程序 (半区间搜寻法) ·····	52
4. 求三角多项式的值计算程序·····	54
5. 多元线性方程组求解计算程序·····	56

第五章 矢量和矩阵计算程序

1. 矢量分析程序·····	58
2. 矢量的加、减、点积、叉积计算程序·····	60
3. 矩阵的加、减、标量积计算程序·····	61
4. 矩阵的矢量积计算程序·····	63
5. 矩阵求逆计算程序·····	65

第六章 数制、单位、坐标转换和绘制函数的图形程序

1. 数制转换之一: 十六进制—十进制转换程序·····	68
2. 数制转换之二: 二进制—十进制相互转换程序·····	70
3. 单位换算之一: 弧度—角度换算程序·····	72
4. 单位换算之二: 角度—弧度换算程序·····	73
5. 单位换算之三: 华氏—摄氏温度换算程序·····	75
6. 单位换算之四: 十七种英制—公制单位换算程序·····	76
7. 坐标转换: 极坐标—笛卡尔坐标系转换程序·····	79
8. 由坐标绘制其图形程序·····	80
9. 绘制函数在极坐标系中的图形程序·····	84
10. 函数的图形绘制程序·····	88
11. 线性插值计算程序·····	90
12. 曲线插值计算程序·····	92

第七章 投资、贷款、储蓄、折旧处理和债务偿还等有关程序

1. 投资的名义利率计算程序	91
2. 投资的实际利率计算程序	96
3. 投资的未来值计算程序	97
4. 初始投资额计算程序	98
5. 由投资中定期退股额计算程序	99
6. 定期退股的最小投资计算程序	101
7. 贷款额计算程序—1	102
8. 贷款额计算程序—2	103
9. 贷款的定期偿还计算程序	105
10. 贷款的最终偿还额计算程序	106
11. 贷款偿还期计算程序	107
12. 贷款的年利率计算程序	109
13. 贷款剩余额计算程序	110
14. 定期储蓄未来值计算程序	112
15. 定期储蓄计算程序之一	113
16. 定期储蓄计算程序之二	111
17. 折旧率计算程序	115
18. 折旧处理价格计算程序	116
19. 总折旧额计算程序	118
20. 赚得利息表计算程序	119
21. 债务偿还表计算程序	122

第八章 线性规划计算程序——单纯形法

第九章 通用报表处理程序

1. 表格符号生成程序	135
2. 汉字显示（或打印）程序	136
3. 汉字打印程序	138
4. 通用空表头打印程序	139
5. 通用空表格打印程序	144
6. 报表标题处理程序	119
7. 通用表头打印程序	154
8. 报表封面设计打印程序	160
9. 报表正文打印程序	165
10. 空表头缩小打印程序	172
11. 实用表头缩小打印程序	177

12. 通用报表打印程序.....	182
-------------------	-----

第十章 其它实用程序

1. 找最大和最小数计算程序.....	191
2. 人口增长计算程序.....	192
3. 按产值分类计算程序.....	193
4. 简易工资表计算打印程序.....	194
5. 查询人员程序.....	196
6. 人造卫星坠落计算程序.....	197
7. 最大公因数计算程序.....	199
8. 因数分解计算程序.....	200
9. 加法器程序.....	201
10. 多边形的面积计算程序.....	202
11. 按字母顺序排序计算程序.....	203
12. 解三角形计算程序.....	205
13. 勘测(测绘)计算程序.....	206
14. 系统可靠性计算程序.....	210
15. 平均销售增长率和预测未来销售额计算程序.....	211
16. 计算X天是星期几程序.....	213
17. 两日期间的总天数计算程序.....	214
18. 中、英文未来年历打印程序.....	217
19. 艺术图案设计显示程序.....	223
20. 艺术图案设计打印程序.....	226
主要参考文献.....	231

第一章 数理统计程序

本章包括两个部分：即（一）数据整理及统计程序。（二）常用统计分布及检验计算程序。共15个程序，约有30K。

在生产管理和科学试验中，经常要接触大量的数据，这些数据提供了很多有用的，有时是很重要的信息。但这些有用信息往往不是一目了然的，而是蕴藏在大量数据之中，必须去粗取精，去伪存真，对数据作出科学的整理和分析才能把有用信息提炼出来。

这些数据有两个重要的特征：（1）具有波动性；（2）具有规律性。如在钢铁工业中，每炉铁水的含炭量；某种产品的废品率；在气象方面，某年某月的降雨量；在纺织工业方面，化纤厂在正常生产条件下维尼纶、绵纶的纤度；在军事方面，研究炮弹的杀伤半径；在电子工业方面，电子产品的可靠性等，所遇到的数据都有此特征。对这些数据的研究和整理以及规律性的寻求以前基本上是靠人工计算，不仅繁杂，而且有时要翻查多种近似的计算表格，费人费力精度也低，本章所提供的15个实用程序中有12个是专门用微型计算机来处理这方面问题的。

下面对主要一些数理统计实用程序的主要使用场合作些介绍。

本章给出了六种重要频数分布计算程序：

其中二项式分布计算程序主要用来进行如产品检验，如某种农药的杀虫效率等。在可靠性研究中，用来计算冗余元件相同的并行工作冗余系统的成功概率，也适用于计算其可靠性依赖于时间的元件、一次使用的设备（如多级导弹分离器等）的成功概率等。

普哇松分布在纺织工业中是经常使用的，在可靠性研究中主要是用来计算后备冗余系统的可靠度。

正态（高斯）分布计算程序在许多自然现象和设备特性的研究中都有应用。

χ^2 分布是用来比较观察频率与假设分布的理论频率的，还可用来进行符合度检验。

学生氏T分布计算程序用来计算如有限子样抽样结果所产生的误差。

F分布计算程序可以帮助进行诸如确定哪个输入特性对所研究的输出有显著影响以及哪些输入特性之间存在相互影响等。

本章中还给出了三个统计检验程序：

统计检验计算程序主要用来处理如随机误差和条件误差以及回归方程的显著性检验等方面的问题，利用其进行符合度检验。其中U检验计算程序主要用于标准差比较稳定，只要对其均值作检验的场合。 χ^2 检验主要用于符合度分析等方面。

在本章中还给出了三个运动员和学生成绩统计计算程序。

1. 二项分布计算程序

程序功能:

已知一个数据总体遵从二项分布, 一个产品 (或试验), 要么合格 (成功), 要么是废品 (失败), 二者必居其一。如果已知正常生产时的合格率为 P , 我们抽查 N 个产品看其中有 X 个是废品的概率 R 是多少。只要回答程序所问的 N , X 和 P 程序即可自动计算出其概率 R 。

程序中使用的主要符号说明:

N : NUMBER OF TRIAL——抽查 (或试验) 数

X : EXACT NUMBER OF SUCCESS——合格 (或成功) 数

P : PROBABILITY OF SUCCESS——已知合格 (或成功) 的概率

程序所依据的理论计算公式:

$$P = C_N^X (1 - P)^X P^{N-X}$$

程序名称: BINOMIAL. DIS

```
5 REM THIS IS COMPUTE BINOMIAL DISTRIBUTION PROGRAM
10 PRINT "BINOMIAL DISTRIBUTION"
20 PRINT
30 DIM M(3)
40 PRINT "(TO END PROGRAM TO ENTER 0)"
50 PRINT "NUMBER OF TRIALS";
60 INPUT N
70 IF N=0 THEN 270
80 PRINT "EXACT NUMBER OF SUCCESSES";
90 INPUT X
100 PRINT "PROBABILITY OF SUCCESS";
110 INPUT P
119 REM COMPUTE THE FACTORIALS
120 M(1)=N
130 M(2)=X
140 M(3)=N-X
150 FOR I=1 TO 3
160 IF M(I)=0 THEN 220
170 A=1
180 FOR J=1 TO M(I)
190 A=A*J
200 NEXT J
210 M(I)=LOG(A)
220 NEXT I
229 REM USING THE COMPUTED FACTORIALS, COMPUTE PROBABILITY
230 R=EXP(M(1)-M(2)-M(3)+X*LOG(P)+(N-X)*LOG(1-P))
240 PRINT "PROBABILITY OF ";X;"SUCCESSES IN";N;"TRIALS =" ;R
250 PRINT
260 GOTO 50
270 END
```

Ready:

RUN

BINOMIAL DISTRIBUTION

(TO END PROGRAM TO ENTER 0)

NUMBER OF TRIALS? 5

EXACT NUMBER OF SUCCESSES? 3

PROBABILITY OF SUCCESS? .5

PROBABILITY OF 3 SUCCESSES IN 5 TRIALS = .3125

NUMBER OF TRIALS? 5

EXACT NUMBER OF SUCCESSES? 2

PROBABILITY OF SUCCESS? .180066667

PROBABILITY OF 2 SUCCESSES IN 5 TRIALS = .18747322935

NUMBER OF TRIALS? 0

Ready:

2. 普哇松 (POISSON) 分布计算程序

程序功能:

已知一个数据样本总体遵从普哇松分布, 程序能够由输入的预测频率值L 和试验频率值X自动计算出事件发生的概率A。普哇松分布常用于纺织以及电子等工业中, 用于计算后备冗余系统的可靠度以及其他一些“稠容性”的问题中。又例如: 在一段时间内候车的旅客数、原子放射粒子数等。

程序中使用的主要符号说明:

POISSON DISTRIBUTION——普哇松分布

L: CALCULATED FREQUENCY——预测频率

X: TEST FREQUENCY——试验频率

A: PROBABILITY OF OCCURENCE——事件发生的概率

程序所依据的理论计算公式:

$$P = \frac{\lambda^m}{m!} e^{-\lambda}$$

程序名称: POISSON. DIS

```
10 REM THIS IS CALCULATE POISSON DISTRIBUTION PROGRAM
20 PRINT "POISSON DISTRIBUTION"
30 PRINT
40 PRINT "(TO END PROGRAM ENTER 0)"
50 PRINT "CALCULATE FREQUENCY";
```

```

60 INPUT L
70 REM END PROGRAM ?
80 IF L=0 THEN 230
90 PRINT "TEST FREQUENCY":
100 INPUT X
110 REM COMPUTE FACTORIAL
120 A=1
130 FOR I=1 TO X
140 A=A*I
150 NEXT I
160 REM COMPUTE PROBABILITY
170 A=LOG(A)
180 A=EXP(-L+X*LOG(L)-A)
190 PRINT "PROBABILITY OF ";X;" OCCURRENCES =" ;A
200 PRINT
210 REM RESTART PROGRAM
220 GOTO 50
230 END

```

Ready:

RUN

POISSON DISTRIBUTION

(TO END PROGRAM ENTER 0)

CALCULATE FREQUENCY? 2

TEST FREQUENCY? 4

PROBABILITY OF 4 OCCURRENCES = .090223522158

CALCULATE FREQUENCY? 2

TEST FREQUENCY? 1

PROBABILITY OF 1 OCCURRENCES = .7087056647

CALCULATE FREQUENCY? 0

Ready:

3. 正态分布计算程序

程序功能:

已知一个数据总体遵从正态分布, 其均值是M, 标准离差是S, 利用本程序能够求得该总体的随机样本X落在某个区间中的概率T。

程序中使用的主要符号说明:

NORMAL DISTRIBUTION——正态分布

S——标准离差

M——均值

程序所依据的理论计算公式:

$$T = \frac{1}{\sqrt{2\pi}S} \int_a^b e^{-\frac{(x-M)^2}{2S^2}}$$

程序名称: NORMAL. DIS

```
5 REM THIS IS COMPUTE NORMAL DISTRIBUTION PROGRAM
10 PRINT "NORMAL DISTRIBUTION"
20 PRINT
30 PRINT "(0=STANDARD, 1=NON-STANDARD)"
40 PRINT "WHICH TYPE OF VARIABLE";
50 INPUT S
60 IF S=0 THEN 120
69 REM LINES 70-110 REQUEST NON-STANDARD VARIABLE DATA
70 PRINT "MEAN";
80 INPUT M
90 PRINT "STANDARD DEVIATION";
100 INPUT S
110 GOTO 130
120 S=1
130 PRINT
140 PRINT "TO END PROGRAM ENTER X=99999"
150 PRINT "X =";
160 INPUT Y
170 IF Y=99999 THEN 290
179 REM ADJUST FOR NON-STANDARD VARIABLES
180 Y=(Y-M)/S
189 REM COMPUTE FREQUENCY (Y COORDINATE)
190 R=EXP(-Y^2/2)/2.5066232746
200 PRINT "FREQUENCY =";R
210 Z=Y
219 REM APPROXIMATE PROBABILITY (AREA UNDER CURVE)
220 Y=1/(1+.33267*ABS(Y))
230 T=.5-R*(.4361836*Y-.1201676*Y^2+.937298*Y^3)
239 REM ADJUST FOR NEGATIVE VARIABLES
240 IF Z>0 THEN 260
250 T=1-T
260 PRINT "PROBABILITY =";T
270 PRINT
280 GOTO 150
290 END
```

Ready:

RUN

NORMAL DISTRIBUTION

```

(0=STANDARD, 1=NON-STANDARD)
WHICH TYPE OF VARIABLE? 1
MEAN? 150
STANDARD DEVIATION? 15

TO END PROGRAM ENTER X=99999
X =? 100
FREQUENCY = .053991074211
PROBABILITY = .47724114449

X =? 130
FREQUENCY = .16401040183
PROBABILITY = .59120210693

X =? 99999

Ready:

```

4. X^2 分布计算程序

程序功能:

X^2 分布主要用于元件、系统等可靠性计算等许多方面，它比二项式分布有更广泛的应用，当然这种分布本身是连续的，但也可象二项式分布那样用于离散的情况。具体来说， X^2 分布是用来比较观察频率与假设分布或理论分布的频率。在可靠性理论中的符合度检验问题就是利用 X^2 分布来确定某一理论分布是否适合用来描述测量数据。本程序能够由输入的自由度值 V 和 X^2 值自动计算出其结尾值 $(1 - J * K * L)$ 。

程序中使用的主要符号说明:

V: DEGREE OF FREEDOM——自由度

W: X ——SQUARE—— X^2

K: THE NUMERATOR PRODUCT——分子乘积

R: DENOMINATOR——分母乘积

程序所依据的理论计算公式:

$$Z = 1 + \sum_{m=1}^{\infty} \frac{(X^2)^m}{(V+2) * (V+4) * \dots * (V+2m)}$$

程序名称: X——SQUARE. DIS

```

5 REM THIS IS CALCULATE X-SQUARE DISTRIBUTION PROGRAM
10 PRINT "X-SQUARE DISTRIBUTION"
20 PRINT
30 PRINT "(TO END PROGRAM ENTER 0)"
40 PRINT "DEGREE OF FREEDOM";

```



```

50 INPUT V
60 IF V=0 THEN 280
70 PRINT "X-SQUARE";
80 INPUT W
89 REM R= THE DENOMINATOR PRODUCT
90 R=1
100 FOR I=V TO 2 STEP -2
110 R=R*I
120 NEXT I
129 REM K= THE NUMERRATOR PRODUCT
130 K=W^(INT((V+1)/2))*EXP(-W/2)/R
139 REM THE PI FACTOR IF USED ONLY WHEN DEG . FREDOM IS ODD
140 IF INT(V/2)=V/2 THEN 170
150 J=SQR(2/W/3.1415926536)
160 GOTO 180
165 REM L (SUMMATION FACTOR) CALCULATED LINES 170-240
170 J=1
180 L=1
190 M=1
200 V=V+2
210 M=M*W/V
219 REM
220 IF M<1E-07 THEN 250
230 L=L+M
240 GOTO 200
250 PRINT "TIAL END VALUE =" ; 1-J*K*L
260. PRINT
270 GOTO 40
280 END

```

Ready:

RUN

X-SQUARE DISTRIBUTION

(TO END PROGRAM ENTER 0)

DEGREE OF FREDOM? 1

X-SQUARE? 2.571108

TIAL END VALUE = .10883148481

DEGREE OF FREDOM? 0

Ready:

5. 学生氏 T 分布计算程序

程序功能:

它可用来计算有限子样抽样结果所产生的误差, 若将估计值的误差同置信度联系起来它还能提供估计均值的置信度, 另外 T 分布的另一直接应用是将子样均值与已知的或假设的母体均值进行比较, 以确定该子样是否取自此母体。